Recovering Holomorphic Functions from Their Real or Imaginary Parts without the Cauchy-Riemann Equations

Mathematica supplement to SIAM Review (paper of the same name) Vol 46, No. 4, 2004, pp 717-718.

William T. Shaw
Update 2.01 Jan 4 2005
shaww@maths.ox.ac.uk

1. Introduction

This note contains updates on the paper in the light of comments and information received, and more *Mathematica* code.

This *Mathematica* code is (as of Nov 17th 2004, an improved form of) the implementation of the algorithms described in Section 6 of the paper (with the same title as above) appearing in the Education section of SIAM Review, Vol 46, No. 4, 2004. Please notify any bugs or suggestions for changes/additions to WTS at the e-mail address provided above. As well as the definitions for extraction $f(z)$ from its imaginary part, this notebook also contains further examples, and some code additional to that described in the paper, that adds a check for harmonicity. A function that computes harmonic conjugates using these methods is also included.

Note: β is always a real arbitrary constant.

Update Notes:
1.1: 11 Nov 2004, Theoretical discussion of uniqess and its statement for real analytic u, v.
1.2: 17 Nov 2004, Modification of *Mathematica* code to deal with perverse representations of u, v (vestigial complex operations) and force input functions to real analytic form prior to complexification.
2.0: 29 Dec 2004, Update on the history in the light of information received.
2.01: 4 Jan 2005, I managed to combine the latest historical information with the old version of the code in 2.0 of 29 Dec. This is now fixed and this notebook has both the latest code and the latest information on the history.

This notebook was prepared with version 5.01. Use of 5.1 appears to give the same results, except that notationally the unwieldy "Conjugate" has been replaced by "*" in output.

2. Updates on the history of the method

History update 3 - Xmas 2004

I have just received an e-mail from Timo Betcke (TB) at the Comlab in Oxford stating that there are interesting links with the work of I.N.Vekua from the 1940s, on the relationships between elliptic equations and holomorphic functions - this may of course take these ideas back before Ahlfors (1953), but I have not yet had the chance to read the references supplied by TB.

History update 2 - Dec 14 2004

I have received a fascinating collection of references from Prof. Harold Boas (HB) that appears to do a definitive job of filling in the gaps since the book by Ahlfors that I cited. The following is my commentary on HB's information, based on the refs I have had chance to read thus far - I will also list the one's I have not read at the end of this sub-section.

First of all HB points out that the first edition of the Ahlfors text, dating from 1953, contains the simple extraction formula, and I have checked this in a library copy where the discussion indeed appears on p. 41 - so my main request now is for information that would take this result back before **Ahlfors in 1953** (see update 3, which arrived as I was preparing update 2!). HB has also asked J. D'Angelo (see update 1 below) about his source and D'Angelo recalls discovering the formula for himself in 1988.

A more significant omission on my part, despite checking a large number of standard texts, is the reference "Invitation to Complex Analysis", by HB's father, R.P Boas and published by Random House in 1987, where (see pp. 158-164, where the rule is stated and proved along with some other short-cut methods.) One of these short-cut methods is particularly interesting from a historical point of view as its development is curiously interwined with the extraction result I discussed. If one already knows $u(x, y)$ and $v(x, y)$ then

$$f(z) = u(z, 0) + i\, v(z, 0)$$

From my point of view this is less interesting, as you do need to know both u and v, and the extraction result allows the construction of f from ONE of u and v and hence the harmonic conjugate of either. However, the historical link is very interesting. Let's call this the "alternative method" for getting $f(z)$. So far as I am aware (and according to E.V. Laitone, see below), this alternative method was first given by L.M. Milne-Thomson in 1937 ("On the relation of an analytic function of z to its real and imaginary parts"), Math Gazette, No. 244, 21 (1937), pp. 228-229, and it appeared in the first edition of Milne-Thomson's "Theoretical Hydrodynamics", published in 1938. It also appeared in the second (1949) and third (1955) editions. In the fourth (1962) and later editions Milne-Thomson dropped the alternative method and replaced it with the algebraic extraction theorem a la Ahlfors, so far as I can tell, without comment (see note 5.32 on pp. 130-131 of the fourth edition.)

The story of Milne-Thomson's contribution is picked up by E.V. Laitone (EVL) in his 1977 paper. "Relation of the Conjugate Harmonic Functions to f(z), (The American Mathematical Monthly, Vol. 84, No. 4 pp. 281-283. Laitone's emphasis is on the alternative method, and he describes the extraction theorem as "a very dubious procedure that utilized the complex conjugate"!

All of these discussions refer to what I called theorem 2.1, where the base point is the origin. Indeed, EVL's note goes on to say that "Milne-Thomson's method (i.e. in his 4th ed) cannot be justified in general since it results in $(x^2 + y^2) = 0$ so that any $f(z)$ containing z^{-n} cannot be obtained from any combination of u and v". (EVL actually used ϕ and ψ.) So the issue, at least in this particular thread of papers, of how to cope with a general base-point was not properly understood in 1977. The EVL is available online at JSTOR - this is where I got it.

The issue of the base point (again, in this thread of papers) was resolved by R.A. Struble (RAS) in 1979 ("Obtaining analytic functions and conjugate harmonic functions", Quarterly of Applied Mathematics, April 1979). RAS gave a proof of what I called Theorem 3.1 based on Taylor series, and made the point that the identities "should be better known than they appear to be". He also made the explicit comment that (in the context of fluid dynamics) "each stream function can be expressed algebraically in terms of the potential function, and vice versa." [So maybe this theorem should be called the Ahlfors-Struble theorem?]

Other references supplied by HB that I have not yet had chance to read are:

Cartan's 1961 book: The elementary theory of analytic functions of one or several complex variables; (the French version is 1961, the english translation is available as a 1995 Dover edition.

Hullgol. R.R., relation of the conjugate harmonic functions to f(z) in cylindrical polar coordinates.

HB also comments that he would not be surprised if the formula turns up in much older works. I share his view that this may well have been known in the 19th century.

■ History update 1 - Nov 2004

I am grateful to Bill Margolis for e-mailing me to let me know that Theorem 2.1, that I (at this stage!) attribute to L.V. Ahlfors (though I first read about it it Spiegel's student text) was also given as an exercise (ex 3 of Chapter III) by J. D'Angelo in his recent monograph "Inequalities in Complex Analysis" (Mathematical Monograph #28, Mathematical Assoication of America, 2002). I will comment further on this when I get hold of a copy.

3. Mathematical Issues: Uniqueness of the Complexification

B.Margolis (BM) has also commented on the fact that I did not spell out the precise conditions under which the complexification U of u is uniquely defined. Ahlfors was careful to restrict attention to functions that are rational in x and y, probably because of this issue. The first thing to point out is that *given* f, the natural complexifications, U, V of u, v are indeed uniquely defined, having being constructed explicitly in Theorems 2.1, 3.1. The issue is whether, for example, given only "some formula" for $u(x, y)$, whether we can uniquely specify its complexification. I did not address this in Section 2.1, but my view on this is that the answer is "yes" once u has been specified in real analytic form, i.e. expressed as a real power series in x and y about the relevant base point, or, more usefully, in terms of real functions that are expressible as such power series. Since any complex function that is holomorphic on an open set containing the base point is then analytic in a neighbourhood of that base point (Taylor's Theorem) then any u can be so expressed, so this is not a restriction at all. However, you do have to write it in real analytic form. (The *Mathematica* code given below now forces this, as of version 1.2 of this note.) I should have made such a comment in Section 2.1 of the paper. If one does not work in this representation one can end up with different complexifications. Of course, if one restricts attention to power series then one can give an alternative direct and constructive proof (see below), but this does not make clear the elegant role of the reflection principle embodied in Theorems 2.1, 3.1.

For example (supplied by BM), if we take $f(z) = z$ in order to motivate $u(x, y) = x$, we can write u in various ways. Let's consider

$$u(x, y) = x \quad (1)$$

$$u(x,\ y) = \frac{1}{2}\,((x + i\,y) + (x - i\,y)) \tag{2}$$

$$u(x,\ y) = \frac{1}{2}\,((x + i\,y) + \text{Conjugate}(x + i\,y)) \tag{3}$$

$$u(x,\ y) = \frac{1}{2}\,((x - i\,y) + \text{Conjugate}(x - i\,y)) \tag{4}$$

The first two work out fine, while (3) and (4) give x and 0 respectively! So it is important that you simplify down to the real analytic form of u. Equivalently one can contrive a zero u to have a non-zero complexification, but not if you limit attention to the real analytic case.

In terms of educational applications, I think it can safely be left to students to boil the function down to the real analytic case using pen and paper methods. I have added a section below on forcing this automatically with *Mathematica*.

This is related to the reason why *Mathematica*'s **`ComplexExpand`** function has to have its target functions set to **`Re`** and **`Im`**, (see below), in order to provide a purely real characterization of u.

■ The Power Series version of Theorem 3.1

Bearing in mind that Theorem 3.1 tells us what we should we aiming for, it is possible to give a now very simple argument based on power series, which shows that we can specify U uniquely, up to an imaginary constant. (This also makes it clear why (1,2) above are right for complexification and (3,4) are wrong.)

So suppose $u(x,\ y)$ is the real part of a holomorphic function $f(z)$ with the property that f is holomorphic in a neighbourhood of $z = a = a_1 + i\,a_2$. Then, by Taylor's Theorem, there is a unique power series

$$f(z) = \sum_{n=0}^{\infty} c_n\,(z - a)^n\ . \tag{5}$$

It follows that we can write:

$$u(x,\ y) = \frac{1}{2} \sum_{n=0}^{\infty} c_n\,(x + i\,y - a)^n + \frac{1}{2} \sum_{n=0}^{\infty} \bar{c}_n\,(x - i\,y - \overline{a})^n\ . \tag{6}$$

We also note that *given a knowledge of* $u(x,\ y)$ *alone*, on the assumption that it is real anaytic about $(a_1,\ a_2)$ and a solution of Laplace's equation, such a representation must exist and be unique. This is because the Laplace condition is just

$$\frac{\partial^2 u}{\partial z\,\partial \bar{z}} = 0 \iff u = g(z) + h(\bar{z}). \tag{7}$$

and reality constrains h to be expressed in terms of the conjugate of g. Then the existence of a series supplies (6) uniquely, apart from the imaginary part of c_0. Now define $U(w_1,\ w_2)$ *uniquely, up to the imaginary part of* c_0, by

$$U(w_1,\ w_2) = \frac{1}{2} \sum_{n=0}^{\infty} c_n\,(w_1 + i\,w_2 - a)^n + \frac{1}{2} \sum_{n=0}^{\infty} \bar{c}_n\,(w_1 - i\,w_2 - \overline{a})^n\ , \tag{8}$$

and observe that, first

$$U(x,\ y) = \frac{1}{2}\sum_{n=0}^{\infty} c_n\,(x + i\,y - a)^n + \frac{1}{2}\sum_{n=0}^{\infty} \bar{c}_n\,(x - i y - \overline{a})^n \ = \ u(x,\ y)\ , \tag{9}$$

and second,

$$U\left(\frac{z+a}{2},\ \frac{z-\overline{a}}{2\,i}\right) = \\ \frac{1}{2}\sum_{n=0}^{\infty} c_n\,(z-a)^n + \frac{1}{2}\sum_{n=0}^{\infty} \bar{c}_n\,(\overline{a}-\overline{a})^n \ = \ \frac{1}{2}\,f(z) \ + \ \frac{1}{2}\,\bar{c}_0 \ = \ \frac{1}{2}\,f(z) \ + \ \frac{1}{2}\,\overline{f(a)} \tag{10}$$

The result then follows.

4. Mathematica Definition of the Extraction Functions

■ Getting the holomorphic function from its real part

Updated November 17th 2004 to force u into explicit real analytic form prior to its complexification. This will deal correctly with perverse formulae for u containing vestigial complex operations.

```
RealToHolo[expr_, anum_, {xsym_, ysym_, zsym_ }] :=
 Module[{abar = Conjugate[anum], exprf},
  exprf = ComplexExpand[expr, TargetFunctions → {Re, Im}];
  func =
   2 * exprf /.
    {xsym → (zsym + abar) / 2, ysym → (zsym - abar) / (2 * I)};
  basecorr = - exprf /. {xsym → Re[anum], ysym → Im[anum]};
  FullSimplify[func + basecorr + I * β]]
```

■ Testing it, as in the paper

```
TestUSet = {x^2 - y^2, 4 x y (y^2 - x^2), Exp[x] Cos[y],
   Exp[x / (x^2 + y^2)] Cos[y / (x^2 + y^2)], 1 / 2 Log[x^2 + y^2],
   (x^2 + y^2) ^ (1 / 4) Cos[1 / 2 ArcTan[x, y]]};
```

```
Map[RealToHolo[#, 1, {x, y, z}] &, TestUSet]
```

$\left\{z^2 + i\,\beta,\ i\,(z^4 + \beta - 1),\ i\,\beta + e^z,\ i\,\beta + e^{\frac{1}{z}},\ i\,\beta + \log(z),\ i\,\beta + \sqrt{z}\right\}$

Let's see if we can induce any non-uniqueness by writing down u in various different ways:

```
PerverseUSet = {x, 1 / 2 ((x + I y) + (x - I y)),
  1 / 2 ((x + I y) + Conjugate[(x + I y)]),
  1 / 2 ((x - I y) + Conjugate[(x - I y)])}
```

$\left\{x, x, \frac{1}{2}(x + i\,y + \text{Conjugate}(x) - i\,\text{Conjugate}(y)), \frac{1}{2}(x - i\,y + \text{Conjugate}(x) + i\,\text{Conjugate}(y))\right\}$

Note: the appearance of the * rather than the long-winded Conjugate is a feature of 5.1. This notebook was prepared in 5.0.

```
Map[RealToHolo[#, 0, {x, y, z}] &, PerverseUSet]
```

$\{z + i\,\beta, z + i\,\beta, z + i\,\beta, z + i\,\beta\}$

■ Getting the holomorphic function from its imaginary part

```
ImToHolo[expr_, anum_, {xsym_, ysym_, zsym_ }] :=
 Module[{abar = Conjugate[anum], exprf},
  exprf = ComplexExpand[expr, TargetFunctions → {Re, Im}];
  func =
   2 * I * exprf /.
    {xsym → (zsym + abar) / 2, ysym → (zsym - abar) / (2 * I)};
  basecorr = -I * exprf /. {xsym → Re[anum], ysym → Im[anum]};
  FullSimplify[func + basecorr + β]]
```

■ Testing it, cf the paper

```
TestVSet = {2*x*y, x^4 + y^4 - 6*x^2*y^2,
   Exp[x]*Sin[y], (-E^(x/(x^2 + y^2)))*
    Sin[y/(x^2 + y^2)], ArcTan[x,y], (x^2+y^2)^(1/4) Sin[1/2
ArcTan[x,y]]};
```

```
Map[ImToHolo[#, 1, {x, y, z}] &, TestVSet]
```

$\left\{z^2 + \beta - 1, i\,z^4 + \beta, \beta + e^z - e, \beta + e^{\frac{1}{z}} - e, \beta + \log(z), \beta + \sqrt{z} - 1\right\}$

5. The Inverse of **ComplexExpand**

The examples given above all start with manifestly real analytic u unless you consider the "perverse" set. A careful student may wish to explore what happens when you supply u not in this form, and it is also the case that u generated by the use of **ComplexExpand** will not necessarily be in real analytic form. Here is a good example:

```
badexpr = Simplify[ComplexExpand[Re[(x + I y)^(1/5)]]]
```

$$\sqrt[10]{x^2 + y^2}\,\cos\left(\frac{1}{5}\,\mathrm{Arg}(x + i\,y)\right)$$

With the version of **RealToHolo** that automatically applies **ComplexExpand**, this gets sorted out.

```
RealToHolo[badexpr, 1, {x, y, z}]
```

$$i\,\beta + \sqrt[5]{z}$$

Settings for ComplexExpand

To deal with such issues and get **ComplexExpand** to supply real analytic forms, we force *Mathematica* to work out the function in purely real terms:

```
SetOptions[ComplexExpand, TargetFunctions → {Re, Im}];
```

Now all is well:

```
goodexpr = Simplify[ComplexExpand[Re[(x + I y)^(1/5)]]]
```

$$\sqrt[10]{x^2 + y^2}\,\cos\left(\frac{1}{5}\,\tan^{-1}(x, y)\right)$$

As noted above, our extraction functions sort this out, but if you want to use **ComplexExpand** to display results in real analytic form, remember to set the target functions.

```
RealToHolo[goodexpr, 1, {x, y, z}]
```

$$i\,\beta + \sqrt[5]{z}$$

```
Simplify[ComplexExpand[Re[1 / (x + I y) ^2]]]
```

$$\frac{x^2 - y^2}{(x^2 + y^2)^2}$$

```
RealToHolo[%, 1, {x, y, z}]
```

$$i\,\beta + \frac{1}{z^2}$$

■ Example 2

```
Simplify[ComplexExpand[Re[1 / (x + I y) ^4]]]
```

$$\frac{x^4 - 6\,y^2\,x^2 + y^4}{(x^2 + y^2)^4}$$

```
RealToHolo[%, 1, {x, y, z}]
```

$$i\,\beta + \frac{1}{z^4}$$

■ Example 3

```
ComplexExpand[Re[Exp[1 / (x + I y)]]]
```

$$e^{\frac{x}{x^2+y^2}}\,\cos\left(\frac{y}{x^2 + y^2}\right)$$

```
RealToHolo[%, 1, {x, y, z}]
```

$$i\,\beta + e^{\frac{1}{z}}$$

■ Example 4

```
Simplify[
 ComplexExpand[Im[(x + I y) ^ (1 / n)], TargetFunctions → {Re, Im}]]
```

$$(x^2 + y^2)^{\frac{1}{2n}} \sin\left(\frac{\tan^{-1}(x, y)}{n}\right)$$

```
ImToHolo[%, 1, {x, y, z}]
```

$$\beta + 2\sqrt{z^{\frac{1}{n}}} \sinh\left(\frac{\log(z)}{2n}\right)$$

With one further kick this can be sorted out.

```
FullSimplify[PowerExpand[%]]
```

$$z^{\frac{1}{n}} + \beta - 1$$

■ Example 5 - example problem

Of what holomorphic function is this following expression the real part?

```
expr = (1 / (x^2 + y^2)^100) *
   (x^100 - 4950 * y^2 * x^98 + 3921225 * y^4 * x^96 -
     1192052400 * y^6 * x^94 + 186087894300 * y^8 * x^92 -
     17310309456440 * y^10 * x^90 + 1050421051106700 * y^12 * x^88 -
     44186942677323600 * y^14 * x^86 + 1345860629046814650 *
      y^16 * x^84 - 30664510802988208300 * y^18 * x^82 +
     535983370403809682970 * y^20 * x^80 - 7332066885177656269200 *
      y^22 * x^78 + 79776075565900368755100 * y^24 * x^76 -
     699574816500972464467800 * y^26 * x^74 +
     4998813702034726525205100 * y^28 * x^72 -
     29372339821610944823963760 * y^30 * x^70 +
     143012501349174257560226775 * y^32 * x^68 -
     580717429720889409486981450 * y^34 * x^66 +
     1977204582144932989443770175 * y^36 * x^64 -
     5670048986634686922786117600 * y^38 * x^62 +
     13746234145802811501267369720 * y^40 * x^60 -
     28258808871162574166368460400 * y^42 * x^58 +
     49378235797073715747364762200 * y^44 * x^56 -
     73470998190814997343905056800 * y^46 * x^54 +
     93206558875049876949581681100 * y^48 * x^52 -
     100891344545564193334812497256 * y^50 * x^50 +
     93206558875049876949581681100 * y^52 * x^48 -
     73470998190814997343905056800 * y^54 * x^46 +
     49378235797073715747364762200 * y^56 * x^44 -
     28258808871162574166368460400 * y^58 * x^42 +
     13746234145802811501267369720 * y^60 * x^40 -
     5670048986634686922786117600 * y^62 * x^38 +
     1977204582144932989443770175 * y^64 * x^36 -
     580717429720889409486981450 * y^66 * x^34 +
     143012501349174257560226775 * y^68 * x^32 -
     29372339821610944823963760 * y^70 * x^30 +
     4998813702034726525205100 * y^72 * x^28 -
     699574816500972464467800 * y^74 * x^26 +
     79776075565900368755100 * y^76 * x^24 -
     7332066885177656269200 * y^78 * x^22 +
     535983370403809682970 * y^80 * x^20 -
     30664510802988208300 * y^82 * x^18 + 1345860629046814650 *
      y^84 * x^16 - 44186942677323600 * y^86 * x^14 +
     1050421051106700 * y^88 * x^12 - 17310309456440 * y^90 * x^10 +
     186087894300 * y^92 * x^8 - 1192052400 * y^94 * x^6 +
     3921225 * y^96 * x^4 - 4950 * y^98 * x^2 + y^100);
```

Method 1 - ask students to do it with pen and paper (very bad idea).

Method 2:

```
RealToHolo[expr, 1, {x, y, z}]
```

$$i\,\beta + \frac{1}{z^{100}}$$

Larger expressions may need kicking with **`Expand`** or **`Simplify`** again.

I should confess that I originally made the *u* of this example with

```
expr = Simplify[ComplexExpand[
   Re[1 / (x + I y) ^100], TargetFunctions → {Re, Im}]]
```

$$\frac{1}{(x^2+y^2)^{100}}\big(x^{100} - 4950\, y^2\, x^{98} + 3921225\, y^4\, x^{96} - 1192052400\, y^6\, x^{94} + 186087894300\, y^8\, x^{92} - 17310309456440\, y^{10}\, x^{90} + 1050421051106700\, y^{12}\, x^{88} - 44186942677323600\, y^{14}\, x^{86} + 1345860629046814650\, y^{16}\, x^{84} - 30664510802988208300\, y^{18}\, x^{82} + 535983370403809682970\, y^{20}\, x^{80} - 7332066885177656269200\, y^{22}\, x^{78} + 79776075565900368755100\, y^{24}\, x^{76} - 699574816500972464467800\, y^{26}\, x^{74} + 4998813702034726525205100\, y^{28}\, x^{72} - 29372339821610944823963760\, y^{30}\, x^{70} + 143012501349174257560226775\, y^{32}\, x^{68} - 580717429720889409486981450\, y^{34}\, x^{66} + 1977204582144932989443770175\, y^{36}\, x^{64} - 5670048986634686922786117600\, y^{38}\, x^{62} + 13746234145802811501267369720\, y^{40}\, x^{60} - 28258808871162574166368460400\, y^{42}\, x^{58} + 49378235797073715747364762200\, y^{44}\, x^{56} - 73470998190814997343905056800\, y^{46}\, x^{54} + 93206558875049876949581681100\, y^{48}\, x^{52} - 100891344545564193334812497256\, y^{50}\, x^{50} + 93206558875049876949581681100\, y^{52}\, x^{48} - 73470998190814997343905056800\, y^{54}\, x^{46} + 49378235797073715747364762200\, y^{56}\, x^{44} - 28258808871162574166368460400\, y^{58}\, x^{42} + 13746234145802811501267369720\, y^{60}\, x^{40} - 5670048986634686922786117600\, y^{62}\, x^{38} + 1977204582144932989443770175\, y^{64}\, x^{36} - 580717429720889409486981450\, y^{66}\, x^{34} + 143012501349174257560226775\, y^{68}\, x^{32} - 29372339821610944823963760\, y^{70}\, x^{30} + 4998813702034726525205100\, y^{72}\, x^{28} - 699574816500972464467800\, y^{74}\, x^{26} + 79776075565900368755100\, y^{76}\, x^{24} - 7332066885177656269200\, y^{78}\, x^{22} + 535983370403809682970\, y^{80}\, x^{20} - 30664510802988208300\, y^{82}\, x^{18} + 1345860629046814650\, y^{84}\, x^{16} - 44186942677323600\, y^{86}\, x^{14} + 1050421051106700\, y^{88}\, x^{12} - 17310309456440\, y^{90}\, x^{10} + 186087894300\, y^{92}\, x^8 - 1192052400\, y^{94}\, x^6 + 3921225\, y^{96}\, x^4 - 4950\, y^{98}\, x^2 + y^{100}\big)$$

■ Adding a check for u or v to be a harmonic function

Note that as defined, neither of the functions described above care about whether the *u* or *v* you supply to them are necessarily the real or imaginary parts of a holomorphic function. They return results whatever you give them! For example.

```
RealToHolo[x^2 + y^2, 0, {x, y, z}]
```

$i\,\beta$

```
RealToHolo[x^2 + y^2, 1, {x, y, z}]
```

$2z + i\,\beta - 1$

Note the dependence on the base point! It is far from clear whether such transformations have any useful interpretation, and they certainly do not serve as an inverse to **ComplexExpand**. For example

```
ComplexExpand[Re[% /. {z → x + I y, β → 0}]]
```

$2x - 1$

So it is a good idea to include a check to exclude this case. In the following extended code we include an explicit check that the Laplacian operator gives zero. Note that this check involves (a) invoking **FullSimplify** to boil down the Laplacian as much as possible, having used **Together** to put things over a common denominator; (b) the use of the "identically equals" within *Mathematica*, given by ===.

```
RealToHoloCheck[expr_, anum_, {xsym_, ysym_, zsym_ }] :=
 Module[{abar = Conjugate[anum], exprf,
   (*This may need improving to stop false negatives *)
   laplacian = FullSimplify[
     Together[D[expr, {xsym, 2}] + D[expr, {ysym, 2}]]]},
  exprf = ComplexExpand[expr, TargetFunctions → {Re, Im}];
  If[laplacian === 0,
   func =
    2 * exprf /.
     {xsym → (zsym + abar) / 2, ysym → (zsym - abar) / (2 * I)};
   basecorr = - exprf /. {xsym → Re[anum], ysym → Im[anum]};
   FullSimplify[func + basecorr + I * β],
   Print["This expression is not the real part of a
      holomorphic function - its Laplacian is "]; laplacian]]
```

```
ImToHoloCheck[expr_, anum_, {xsym_, ysym_, zsym_ }] :=
 Module[{abar = Conjugate[anum], exprf,
   (*This may need improving to stop false negatives *)
   laplacian = FullSimplify[
     Together[D[expr, {xsym, 2}] + D[expr, {ysym, 2}]]]},
  exprf = ComplexExpand[expr, TargetFunctions → {Re, Im}];
  If[laplacian === 0,
   func =
    2 * I * exprf /.
     {xsym → (zsym + abar) / 2, ysym → (zsym - abar) / (2 * I)};
   basecorr =  -I * exprf /. {xsym → Re[anum], ysym → Im[anum]};
   FullSimplify[func + basecorr + β],
   Print["This expression is not the imaginary part of a
      holomorphic function - its Laplacian is "]; laplacian]]
```

Note that the output is the Laplacian if the check fails, so you do at least have the chance to do further manual simplifications of the Laplacian if you really think the input is harmonic, and then use the non-checking code to finalize matters.

```
RealToHoloCheck[x^2 + y^2, 1, {x, y, z}]
```

This expression is not the real part of a holomorphic function − its Laplacian is

4

```
ImToHoloCheck[(x^2 - y^2)^4, 1, {x, y, z}]
```

This expression is not the imaginary part of a holomorphic function − its Laplacian is

$48(x^2 - y^2)^2 (x^2 + y^2)$

```
Map[RealToHoloCheck[#, 1, {x, y, z}] &, TestUSet]
```

$\{z^2 + i\beta, i(z^4 + \beta - 1), i\beta + e^z, i\beta + e^{\frac{1}{z}}, i\beta + \log(z), i\beta + \sqrt{z}\}$

```
Map[ImToHoloCheck[#, 1, {x, y, z}] &, TestVSet]
```

$\{z^2 + \beta - 1, i z^4 + \beta, \beta + e^z - e, \beta + e^{\frac{1}{z}} - e, \beta + \log(z), \beta + \sqrt{z} - 1\}$

What about example 5? Now checking the Laplacian takes a little while.

```
RealToHoloCheck[expr, 1, {x, y, z}]
```

$i\,\beta + \frac{1}{z^{100}}$

Comments on how to improve these functions is very welcome, especially if you find cases where the Laplacian check fails!

6. Harmonic Conjugates

This is now a matter of sticking together our new function (now with the check) with **ComplexExpand.** We do not need to introduce a complex variable here in the arguments, so it is just used internally.

```
SetOptions[ComplexExpand, TargetFunctions → {Re, Im}];
```

```
HarmonicConjugate[expr_, anum_, {xsym_, ysym_}] :=
 Module[
  {abar = Conjugate[anum], zsym, exprf, laplacian = FullSimplify[
     Together[D[expr, {xsym, 2}] + D[expr, {ysym, 2}]]]},
  exprf = ComplexExpand[expr, TargetFunctions → {Re, Im}];
  If[laplacian === 0,
   func =
    2 * exprf /.
     {xsym → (zsym + abar) / 2, ysym → (zsym - abar) / (2 * I)};
   basecorr = - exprf /. {xsym → Re[anum], ysym → Im[anum]};
   ComplexExpand[Im[FullSimplify[
      (func + basecorr + I * β) /. zsym → xsym + I * ysym]]],
   Print["This expression is not the real part of a
      holomorphic function - its Laplacian is "]; laplacian]]
```

```
ComplexExpand[Re[(x + I y) ^ 4]]
```

$x^4 - 6\,y^2\,x^2 + y^4$

```
ComplexExpand[Im[(x + I y) ^ 4]]
```

$4\,x^3\,y - 4\,x\,y^3$

```
HarmonicConjugate[p * (x^4 - 6 x^2 y^2 + y^4), 0, {x, y}]
```

$4\,p\,y\,x^3 - 4\,p\,y^3\,x + \beta$

```
TraditionalForm[TestUSet]
```

$$\left\{x^2 - y^2,\ 4\,x\,y\,(y^2 - x^2),\ e^x\cos(y),\ e^{\frac{x}{x^2+y^2}}\cos\left(\frac{y}{x^2+y^2}\right),\ \frac{1}{2}\log(x^2+y^2),\ \sqrt[4]{x^2+y^2}\,\cos\left(\frac{1}{2}\tan^{-1}(x,\,y)\right)\right\}$$

```
Map[HarmonicConjugate[#, 1, {x, y}] &, TestUSet]
```

$$\left\{2\,x\,y + \beta,\ x^4 - 6\,y^2\,x^2 + y^4 + \beta - 1,\ \beta + e^x\sin(y),\ \beta - e^{\frac{x}{x^2+y^2}}\sin\left(\frac{y}{x^2+y^2}\right),\ \beta + \tan^{-1}(x,\,y),\ \beta + \sqrt[4]{x^2+y^2}\,\sin\left(\frac{1}{2}\tan^{-1}(x,\,y)\right)\right\}$$

```
TableForm[Transpose[{%%, %}]]
```

$x^2 - y^2$	$2\,x\,y + \beta$
$4\,x\,y\,(y^2 - x^2)$	$x^4 - 6\,y^2\,x^2 + y^4 + \beta - 1$
$e^x\cos(y)$	$\beta + e^x\sin(y)$
$e^{\frac{x}{x^2+y^2}}\cos\left(\frac{y}{x^2+y^2}\right)$	$\beta - e^{\frac{x}{x^2+y^2}}\sin\left(\frac{y}{x^2+y^2}\right)$
$\frac{1}{2}\log(x^2+y^2)$	$\beta + \tan^{-1}(x,\,y)$
$\sqrt[4]{x^2+y^2}\,\cos(\frac{1}{2}\tan^{-1}(x,\,y))$	$\beta + \sqrt[4]{x^2+y^2}\,\sin(\frac{1}{2}\tan^{-1}(x,\,y))$

You can use other variables if you want

```
HarmonicConjugate[a^4 + b^4 - 6 a^2 b^2, 0, {a, b}]
```

$4\,b\,a^3 - 4\,b^3\,a + \beta$

But you will not get anywhere if the function is not harmonic!

```
HarmonicConjugate[a^4 + b^4, 0, {a, b}]
```

This expression is not the real part of a holomorphic function − its Laplacian is

$12\,(a^2 + b^2)$